
Overview

Esta charla explica por que necesitamos intervalos de
confianza simultáneos y presenta el comando sci que los
calcula en Stata

Esta charla

explica por que los intervalos de confianza en las tablas de
resultados no son útiles cuando hay más que un parámetro
de interés

da una introducción a métodos de pruebas múltiples
presenta el comando sci que calcula el valor crítico
corecto y intervalos de confianza corectos cuando hay más
que un parámetro

Comenzamos con un ejemplo de datos simulados
El modelo verdadero es

y = 1 + β1x1 + β2x2 + β3x3 + ε

cuando β1 = β2 = β3 = 0
ε es independiente de cada xj , yx1

x2

x3

 ∼ N

0
0
0

 ,

 1 −.4 −.4
−.4 1 −.4
−.4 −.4 1



Aquí son los resultado obtenidos de un solo sorteo del
proceso que genera los data (PGD)

. clear all

. set seed 12343

. local n = 500

. matrix V = (1, -.4, -.4\ -.4, 1, -.4 \ -.4, -.4, 1)

. drawnorm x1 x2 x3, cov(V) n(`n´)
(obs 500)
. generate y = 1 + (rchi2(20)-20)/sqrt(2*20)

. regress y x1 x2 x3
Source SS df MS Number of obs = 500

F(3, 496) = 3.13
Model 7.81247109 3 2.60415703 Prob > F = 0.0253

Residual 412.095425 496 .830837551 R-squared = 0.0186
Adj R-squared = 0.0127

Total 419.907896 499 .84149879 Root MSE = .9115

y Coefficient Std. err. t P>|t| [95% conf. interval]

x1 -.0554501 .0595192 -0.93 0.352 -.172391 .0614908
x2 -.0711631 .0588022 -1.21 0.227 -.1866953 .0443691
x3 -.1684236 .0610912 -2.76 0.006 -.2884531 -.048394

_cons .9518185 .0408308 23.31 0.000 .8715959 1.032041

Ahora, quiero hacer esta prueba H0 : β3 = 0
Ha : β3 6= 0 al nivel de

significancia de 5%
Rechazo la H0, aunque H0 es cierta
El nivel de significancia una tasa de error
El nivel de significancia es la probabilidad que rechazo la
H0, cuando H0 es cierta

Podemos correr una simulación para entender mejor esta
tasa de error
La simulación imita la estructura estadística frequentista
Fijamos el nivel de significancia (la tasa de error) a 5%
en la simulación,

se saca muchos sorteos del PGD (β3 = 0 )
Usando los datos de cado sorteo

se corre
regress y x1 x2 x3
se defina la variable indicator r a ser 1, si el valor p de
test _b[x3] = 0
es menor que .05 ( r es 0 de lo contrario)

La media de la muestra de r, usando los resultado de los
mucho sorteos, es la proporción de la muestra por el evento
de recharzar H0 caundo la H0 es cierta
La media de la muestra de r debe ser cerca de .05, lo cual
es el nivel de significancia que usamos el las pruebas

. postfile buf rep r using sim1, replace

. forvalues j=1/2000 {
2. quietly drop _all
3. quietly drawnorm x1 x2 x3, cov(V) n(`n´)
4. quietly generate y = 1 + (rchi2(20)-20)/sqrt(2*20)
5. quietly regress y x1 x2 x3
6. quietly test _b[x1] = 0
7. local r = (r(p)<.05)
8. post buf (`j´) (`r´)
9. }

. postclose buf

. use sim1, clear

. summarize r
Variable Obs Mean Std. dev. Min Max

r 2,000 .047 .2116918 0 1

Como esperamos, rechazamos el H0 con un frequencia de
casi 5%, aunque la H0 es cierta

. regress y x1 x2 x3
Source SS df MS Number of obs = 500

F(3, 496) = 3.13
Model 7.81247109 3 2.60415703 Prob > F = 0.0253

Residual 412.095425 496 .830837551 R-squared = 0.0186
Adj R-squared = 0.0127

Total 419.907896 499 .84149879 Root MSE = .9115

y Coefficient Std. err. t P>|t| [95% conf. interval]

x1 -.0554501 .0595192 -0.93 0.352 -.172391 .0614908
x2 -.0711631 .0588022 -1.21 0.227 -.1866953 .0443691
x3 -.1684236 .0610912 -2.76 0.006 -.2884531 -.048394

_cons .9518185 .0408308 23.31 0.000 .8715959 1.032041

Ahora, quiero hacer la prueba
H0 : β1 = β2 = β3 = 0
Ha : max{|β1|, |β2|, |β3|} > 0

y quiero una tasa de error de 5%
En palabras, estoy revisando la hipótesis nula de que todos
los coeficientes son zeros versus la hipótesis alternativa de
que por lo menos uno de los coeficientes no es zero
Estas hipóteses son frequentes cuando uno está
investigado si uno de las variables x1, x2, u x3 tiene algun
efecto

Ahora, quiero hacer la prueba
H0 : β1 = β2 = β3 = 0
Ha : max{|β1|, |β2|, |β3|} > 0

y quiero una tasa de error de 5%
Esta hipótesis nula y esta hipótesis alternativa se puede ver
como una familia de tres pruebas
H0,1 : β1 = 0 H0,2 : β2 = 0 H0,3 : β3 = 0
Rechaza H0 si se rechaza H0,j : βj = 0 por caulquier
j ∈ {1,2,3}
La tasa de error por la famila de pruebas se llama la tasa
de error familiar (TEF)

Usamos αFW por la TEF
La TEF en mayor que el nivel de significancia por cada
prueba en la familia
Si

se rechaza la hipótesis H0 familiar cuando se rechaza
cualqier H0,j en la familia (usando el nivel de significania α),

incurrirá un tasa de error mayor que α

. regress y x1 x2 x3
Source SS df MS Number of obs = 500

F(3, 496) = 3.13

H0 : β1 = β2 = β3 = 0
Ha : max{|β1|, |β2|, |β3|} > 0

La tabla de resulados de los comandos de Stata nos invita usar
el nivel de significancia como fuera la TEF
como en el siguiente procedimiento
SC Procedimiento:

Use el nivel de significancia α como fuera la TEF
rechaze H0 familiar, si el valor p en la tabla de resultados es
menor que α por cualquier de los coeficientes

Es probable que se acuerde de un aviso en contra de este
procidimiento, pero la tabla de resultados te invita usarlo
Corremos una simulación para ver si importa
Puse corelaciones negativas entre las variable x para bajar
el el efecto de hacer mas que una prueba

. postfile buf rep r1 r2 r3 using sim_sc, replace

. forvalues j=1/2000 {
2. quietly drop _all
3. quietly drawnorm x1 x2 x3, cov(V) n(`n´)
4. quietly generate y = 1 + (rchi2(20)-20)/sqrt(2*20)
5. quietly regress y x1 x2 x3
6. quietly test _b[x1] = 0
7. local r1 = (r(p)<.05)
8. quietly test _b[x2] = 0
9. local r2 = (r(p)<.05)
10. quietly test _b[x3] = 0
11. local r3 = (r(p)<.05)
12. post buf (`j´) (`r1´) (`r2´) (`r3´)
13. }

. postclose buf

. use sim_sc, clear

. generate r = max(r1, r2, r3)

. summarize r1 r2 r3 r
Variable Obs Mean Std. dev. Min Max

r1 2,000 .049 .215922 0 1
r2 2,000 .0455 .2084502 0 1
r3 2,000 .051 .2200527 0 1
r 2,000 .107 .3091906 0 1

La TEF en la muestra es 10.70%, en lugar de 5%
La tasa de rechazar cada prueba individual es cerca de 5%

Bonferroni

El método Bonferroni es una manera sencillo y general
para corregir el problema
El método Bonferroni utiliza un límite superior para la TEF
en lugar la TEF actual

Nos aseguramos de que la TEF real sea menor que el que
especificamos

La potencia de una prueba es la probabilidad de que
rechace correctamente H0, cuando H0 es falso
El método Bonferroni puede tener mucha menos potencia
que un procedimiento que utiliza la TEF real en lugar del
límite superior, porque el el límite superior puede ser
mucho mayor que la TEF real

Bonferroni method

El límite superior Bonferroni es sencillo

Si hay m pruebas en la familia y el nivel de significancia de
cada prueba es α, el límite superior de Bonferroni es α

m

El método Bonferroni es general, no limita la dependencia
entre las pruebas en la familia
El método Bonferroni desperdicia potencia

El límite puede ser mucho más grande que la TEF real y, por
lo tanto, tiene una potencia innecesariamente baja

. local b = .05/3

. postfile buf rep r1 r2 r3 using sim_bon, replace

. forvalues j=1/2000 {
2. quietly drop _all
3. quietly drawnorm x1 x2 x3, cov(V) n(`n´)
4. quietly generate y = 1 + (rchi2(20)-20)/sqrt(2*20)
5. quietly regress y x1 x2 x3
6. quietly test _b[x1] = 0
7. local r1 = (r(p)<`b´)
8. quietly test _b[x2] = 0
9. local r2 = (r(p)<`b´)
10. quietly test _b[x3] = 0
11. local r3 = (r(p)<`b´)
12. post buf (`j´) (`r1´) (`r2´) (`r3´)
13. }

. postclose buf

. use sim_bon, clear

. generate r = max(r1, r2, r3)

. summarize r1 r2 r3 r
Variable Obs Mean Std. dev. Min Max

r1 2,000 .0145 .1195696 0 1
r2 2,000 .013 .1133023 0 1
r3 2,000 .0125 .1111302 0 1
r 2,000 .0315 .1747084 0 1

La TEF en la muestra es 3.15% en lugar de 5%
La TEF actual es menor de lo que debe ser por casi 37%
=100(5-3.15)/5

sup Wald/sup t

Un procidimiento de prueba nominal tiene un tasa de
error igual a la tasa de error especificada
Queremos un procidimiento nominal por las pruebas
multiples en la familia
El método sup-Wald/sup-t para calcular le valor crítico
produce un procidimiento nominal

El método sup-Wald/sup-t para calcular le valor crítico se
descubrío antes de que era possible computarlo
Vea (Miller, 1966, capítulo 1) y (Savin, 1984, página 833)

sup Wald/sup t

Para que sea más sensillo, otra vez consideramos
H0 : β1 = β2 = β3 = 0
Ha : max{|β1|, |β2|, |β3|} > 0

El valor crítico ct induce una TEF de αFW , si lo resuelva

Pr
(
max

{
|w1|, |w2|, |w3|

}
≤ ct

)
= αFW

cuando wj = β̂j/sej es la estastica Wald u (t ) de H0,j

En la práctica, podemos simular un gran número de sorteos
del wj = β̂1/sej , porque tenemos un estimador consistente
de su distribución conjunta
Usando estos sorteos, ct es el cuantil de muestra αFW de
estos sorteos
Vea Montiel Olea and Plagborg-Møller (2019) y Drukker
(2021)

El comando sci
funciona después de todos los comandos frecuentistas que
tienen una distribución normal de muestra grande u una
distribución t de muestra pequeña
Calcula el valor crítico ct
Calcula una banda de confianza simulatanea para los
parámetros de interés
Hace los cálculos solo por las variables de interés

Vea Drukker (2021)

Ejemplo después de regress

. regress y x1 x2 x3
Source SS df MS Number of obs = 500

F(3, 496) = 3.13
Model 7.81247109 3 2.60415703 Prob > F = 0.0253

Residual 412.095425 496 .830837551 R-squared = 0.0186
Adj R-squared = 0.0127

Total 419.907896 499 .84149879 Root MSE = .9115

y Coefficient Std. err. t P>|t| [95% conf. interval]

x1 -.0554501 .0595192 -0.93 0.352 -.172391 .0614908
x2 -.0711631 .0588022 -1.21 0.227 -.1866953 .0443691
x3 -.1684236 .0610912 -2.76 0.006 -.2884531 -.048394

_cons .9518185 .0408308 23.31 0.000 .8715959 1.032041

. sci
Sup-Wald simultaneous confidence band
Critical value c = 2.442

y Coef. Std. Err. t [95% Conf. Band]

x1 -.0554501 .0595192 -0.93 -.2007956 .0898954
x2 -.0711631 .0588022 -1.21 -.2147577 .0724315
x3 -.1684236 .0610912 -2.76 -.3176079 -.0192392

_cons .9518185 .0408308 23.31 .85211 1.051527

. display "Single-comparison critical value is " invt(496, .975)
Single-comparison critical value is 1.9647583

Variables de interés
. quietly regress y x1 x2 x3
. sci , pnames(x1 x2 x3)
Sup-Wald simultaneous confidence band
Critical value c = 2.307

y Coef. Std. Err. t [95% Conf. Band]

x1 -.0554501 .0595192 -0.93 -.1927612 .081861
x2 -.0711631 .0588022 -1.21 -.2068201 .0644939
x3 -.1684236 .0610912 -2.76 -.3093613 -.0274858

. sci , pnames(x3)
Sup-Wald simultaneous confidence band
Critical value c = 1.962

y Coef. Std. Err. t [95% Conf. Band]

x3 -.1684236 .0610912 -2.76 -.28826 -.0485872

. display "Single-comparison critical value is " invt(496, .975)
Single-comparison critical value is 1.9647583

20 / 30



. postfile buf rep r1_sci r2_sci r3_sci r1_bon r2_bon r3_bon using sim_sci, rep
> lace
. local c_bon = invt(`n´-4, 1-.05/(2*3))
. forvalues j=1/4000 {
2. quietly drop _all
3. quietly drawnorm x1 x2 x3, cov(V) n(`n´)
4. quietly generate y = 1 + (rchi2(20)-20)/sqrt(2*20)
5. quietly regress y x1 x2 x3
6. qui sci, pnames(x1 x2 x3)
7. local ct = r(c)
8. local w1 = _b[x1]/_se[x1]
9. local w2 = _b[x2]/_se[x2]
10. local w3 = _b[x3]/_se[x3]
11. local r1_sci = (abs(`w1´) > `ct´)
12. local r2_sci = (abs(`w2´) > `ct´)
13. local r3_sci = (abs(`w3´) > `ct´)
14. local r1_bon = (abs(`w1´) > `c_bon´)
15. local r2_bon = (abs(`w2´) > `c_bon´)
16. local r3_bon = (abs(`w3´) > `c_bon´)
17. post buf (`j´) (`r1_sci´) (`r2_sci´) (`r3_sci´) ///

> (`r1_bon´) (`r2_bon´) (`r3_bon´)
18.
. }
. postclose buf

. use sim_sci, clear

. generate r_sci = max(r1_sci, r2_sci, r3_sci)

. generate r_bon = max(r1_bon, r2_bon, r3_bon)

. summarize r1_sci r2_sci r3_sci r_sci r1_bon r2_bon r3_bon r_bon, sep(4)
Variable Obs Mean Std. dev. Min Max

r1_sci 4,000 .02175 .1458844 0 1
r2_sci 4,000 .021 .143402 0 1
r3_sci 4,000 .02425 .1538436 0 1
r_sci 4,000 .05225 .2225586 0 1

r1_bon 4,000 .01825 .1338709 0 1
r2_bon 4,000 .01675 .1283494 0 1
r3_bon 4,000 .0195 .1382915 0 1
r_bon 4,000 .042 .2006142 0 1

. proportion r_sci r_bon
Proportion estimation Number of obs = 4,000

Logit
Proportion Std. err. [95% conf. interval]

r_sci
0 .94775 .0035185 .9404055 .9542334
1 .05225 .0035185 .0457666 .0595945

r_bon
0 .958 .0031716 .9513226 .9637963
1 .042 .0031716 .0362037 .0486774

. sci , pnames(1.r_sci 1.r_bon)
Sup-Wald simultaneous confidence band
Critical value c = 2.112

Proportion Coef. Std. Err. t [95% Conf. Band]

1.r_sci .05225 .0035185 14.85 .0448188 .0596812
1.r_bon .042 .0031716 13.24 .0353015 .0486985

La TEF simulada usando sci es 5.2%, que está cerca la
TEF especificada de 5%
La TEF simulada usando el Bonferroni crítico es 4.2%

. quietly regress y x1 x2 x3

. sci , pnames(x1 x2 x3)
Sup-Wald simultaneous confidence band
Critical value c = 2.310

y Coef. Std. Err. t [95% Conf. Band]

x1 -.0554501 .0595192 -0.93 -.1929681 .0820679
x2 -.0711631 .0588022 -1.21 -.2070245 .0646984
x3 -.1684236 .0610912 -2.76 -.3095737 -.0272734

. display "Single-comparison critical value is " invt(496, .975)
Single-comparison critical value is 1.9647583

El procidimiento sup-Wald produce intervalos de confianza
rectangulares, que son fáciles de interpretar
La banda de confianza (el límite de confianza de
visualización) es para los 3 parámetros simultáneamente
Rechazamos, al nivel del 5%, cualquier vector de valores
(β̃1, β̃2, β̃3) si β̃j está fuera del j (ésimo) componente
estimado de la banda de confianza
Rechazamos, al nivel del 5%, cualquier valor β̃j para el cual
β̃j está fuera del j (ésimo) componente estimado de la
banda de confianza

Otra forma de revisar la prueba de familia H0 es usar test
prueba (_b[x1] = 0) (_b[x2] = 0) (_b[x3] =
0)

Esta prueba conjunta de Wald produce una tasa de error
nominal
Los intervalos de confianza de esta prueba conjunta de
Wald son elípticos, y son muy difíciles de interpretar en
dimensiones superiores

Potencia

La potencia de una prueba es la probabilidad de que
rechace correctamente H0, cuando H0 es falso
A medida que aumenta el tamaño de la muestra, la
potencia del método sup-Wald llega a uno
En muestras finitas, cuando la hipótesis alternativa
(verdadera) no es demasiado diferente de la (falsa)
hipótesis nula, la prueba conjunta de Wald y la prueba
sup-Wald tienen diferentes regiones de rechazo y
diferentes niveles de potencia

Potencia

Drukker (2021) presenta evidencia de simulación que
compara la potencia de la prueba Sup-Wald, la prueba
conjunta-Wald y la prueba utilizando los valores críticos de
Bonferroni
Como se esperaba, según Savin (1984), se puede crear
una alternativa en que el Sup-Wald tiene mayor potencia de
muestra finita que la prueba conjunta de Wald
y
se puede crear una alternativa en que el Sup-Wald tiene
una potencia de muestra finita menor que la prueba
conjunta de Wald
En los resultados de Drukker (2021), hay diseños en los
que la potencia de muestra finita de la prueba conjunta de
Wald es menor que la muestra finita poder del método
Bonferroni

El enfoque sup-Wald es un método de un solo paso
Los métodos de un solo paso pueden producir bandas de
confianza para parámetros
Hay métodos de varios pasos que ofrecen mayor potencia
en el costo de los intervalos de confianza

Vea Romano and Wolf (2005), Romano, Shaikh, and Wolf
(2010), y Clarke, Romano, and Wolf (2020)

Drukker (2021) tiene más ejemplos y más detalles
sci funciona después de cualquier estimador frecuentista
en Stata que implementa un estimador con una distribución
normal de muestra grande o una distribución t de muestra
pequeña
Drukker (2021) analiza una simulación que utiliza efectos
fijos QML Poisson

Conclusión

La tabla de resultados estándar no produce una inferencia
válida para múltiples parámetros
sci produce una inferencia válida y nominal para múltiples
parámetros
sci también es fácil de usar

30 / 30



References

Clarke, D., J. P. Romano, and M. Wolf. 2020. The Romano–Wolf
multiple-hypothesis correction in Stata. The Stata Journal
20(4): 812–843.

Drukker, D. M. 2021. Simultaneous confidence bands for Stata
estimation commands. Under review at The Stata Journal .
URL https://www.researchgate.net/publication/
355165242_Simultaneous_confidence_bands_for_
Stata_estimation_commands.

Miller, R. G. J. 1966. Simultaneous statistical inference.
McGraw-Hill.

Montiel Olea, J. L., and M. Plagborg-Møller. 2019. Simultaneous
confidence bands: Theory, implementation, and an application
to SVARs. Journal of Applied Econometrics 34: 1–17.

Romano, J. P., A. M. Shaikh, and M. Wolf. 2010. Hypothesis
testing in econometrics. Annual Review Economics 2(1):
75–104.

30 / 30

https://www.researchgate.net/publication/355165242_Simultaneous_confidence_bands_for_Stata_estimation_commands.
https://www.researchgate.net/publication/355165242_Simultaneous_confidence_bands_for_Stata_estimation_commands.
https://www.researchgate.net/publication/355165242_Simultaneous_confidence_bands_for_Stata_estimation_commands.


Bibliography

Romano, J. P., and M. Wolf. 2005. Exact and approximate
stepdown methods for multiple hypothesis testing. Journal of
the American Statistical Association 100(469): 94–108.

Savin, N. E. 1984. Multiple hypothesis testing. Handbook of
econometrics 2: 827–879.

30 / 30


	Bibliography

